NAG Library Routine Document

G05PAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05PAF generates a realization of a univariate time series from an autoregressive moving average (ARMA) model. The realization may be continued or a new realization generated at subsequent calls to G05PAF.

2 Specification

```
 SUBROUTINE G05PAF(MODE, XMEAN, IP, PHI, IQ, THETA, AVAR, VAR, N, X,
1                  IGEN, ISEED, R, NR, IFAIL)
 INTEGER           MODE, IP, IQ, N, IGEN, ISEED(4), NR, IFAIL
 double precision  XMEAN, PHI(IP), THETA(IQ), AVAR, VAR, X(N), R(NR)
```

3 Description

Let the vector x_t, denote a time series which is assumed to follow an autoregressive moving average (ARMA) model of the form:

$$x_t - \mu = \phi_1(x_{t-1} - \mu) + \phi_2(x_{t-2} - \mu) + \cdots + \phi_p(x_{t-p} - \mu) + \epsilon_t - \theta_1\epsilon_{t-1} - \theta_2\epsilon_{t-2} - \cdots - \theta_q\epsilon_{t-q} \quad (1)$$

where ϵ_t, is a residual series of independent random perturbations assumed to be Normally distributed with zero mean and variance σ^2. The parameters $\{\phi_i\}$, for $i = 1, 2, \ldots, p$, are called the autoregressive (AR) parameters, and $\{\theta_j\}$, for $j = 1, 2, \ldots, q$, the moving average (MA) parameters. The parameters in the model are thus the p ϕ values, the q θ values, the mean μ and the residual variance σ^2.

G05PAF sets up a reference vector containing initial values corresponding to a stationary position using the method described in Tunnicliffe–Wilson (1979). The routine can then return a realization of $x_1, x_2, \ldots, x_n$. On a successful exit, the recent history is updated and saved in the reference vector R so that G05PAF may be called again to generate a realization of $x_{n+1}, x_{n+2}, \ldots$,etc. See the description of the parameter MODE in Section 5 for details.

One of the initialization routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05PAF.

4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

Tunnicliffe–Wilson G (1979) Some efficient computational procedures for high order ARMA models *J. Statist. Comput. Simulation* **8** 301–309

5 Parameters

1: MODE – INTEGER *Input*

On entry: a code for selecting the operation to be performed by the routine.

MODE = 0
Set up reference vector only.

MODE = 1
Generate terms in the time series using reference vector set up in a prior call to G05PAF.

MODE = 2
Set up reference vector and generate terms in the time series.

Constraint: MODE = 0, 1 or 2.

2: XMEAN – ***double precision*** *Input*

On entry: the mean of the time series.

3: IP – INTEGER *Input*

On entry: p, the number of autoregressive coefficients supplied.

Constraint: IP ≥ 0.

4: PHI(IP) – ***double precision*** array *Input*

On entry: the autoregressive coefficients of the model, $\phi_1, \phi_2, \ldots, \phi_p$.

5: IQ – INTEGER *Input*

On entry: the dimension of the array THETA as declared in the (sub)program from which G05PAF is called. q, the number of moving average coefficients supplied.

Constraint: IQ ≥ 0.

6: THETA(IQ) – ***double precision*** array *Input*

On entry: the moving average coefficients of the model, $\theta_1, \theta_2, \ldots, \theta_q$.

7: AVAR – ***double precision*** *Input*

On entry: σ^2, the variance of the Normal perturbations.

Constraint: AVAR ≥ 0.0.

8: VAR – ***double precision*** *Output*

On exit: the proportion of the variance of a term in the series that is due to the moving-average (error) terms in the model. The smaller this is, the nearer is the model to non-stationarity.

9: N – INTEGER *Input*

On entry: n, the number of observations to be generated.

Constraint: N ≥ 0.

10: X(N) – ***double precision*** array *Output*

On exit: contains the next n observations from the time series.

11: IGEN – INTEGER *Input*

On entry: must contain the identification number for the generator to be used to return a pseudorandom number and should remain unchanged following initialization by a prior call to G05KBF or G05KCF.

12: ISEED(4) – INTEGER array *Communication Array*

On entry: contains values which define the current state of the selected generator.

On exit: contains updated values defining the new state of the selected generator.

13: R(NR) – ***double precision*** array *Communication Array*

On entry: if MODE = 1, the reference vector from the previous call to G05PAF.

On exit: the reference vector.

14: NR – INTEGER *Input*

On entry: the dimension of the array R as declared in the (sub)program from which G05PAF is called.

Constraint: NR $\geq$ IP + IQ + 5 + max(IP, IQ + 1).

15: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, N < 0.

IFAIL = 2

On entry, IP < 0.

IFAIL = 3

On entry, IQ < 0.

IFAIL = 4

On entry, AVAR < 0.

IFAIL = 5

On entry, MODE $\neq$ 0, 1 or 2.

IFAIL = 6

On entry, NR < IP + IQ + 5 + max(IP, IQ + 1).

IFAIL = 7

PHI does not define a stationary autoregressive process.

IFAIL = 8

Either R has been corrupted or the value of IP or IQ is not the same as when R was set up in a previous call with MODE = 0 or 2.

7 Accuracy

The errors in the initialization process should be very much smaller than the error term; see Tunnicliffe–Wilson (1979).

8 Further Comments

The time taken by G05PAF is essentially of order $(\mathrm{IP})^2$.

Note: G05KBF and G05KCF must be used with care if this routine is used as well. The reference vector, as mentioned before, contains a copy of the recent history of the series. This will not be altered properly by calls to any of the above routines. A call to G05KBF or G05KCF should be followed by a call to G05PAF with MODE = 0 to re-initialize the time series reference vector in use. To maintain repeatability with G05KBF, the calls to G05PAF should be performed in the same order and at the same point or points in the simulation every time G05KBF is used. When the generator state is saved and restored using the parameters IGEN and ISEED, the time series reference vector must be saved and restored as well.

The ARMA model for a time series can also be written as:

$$(x_n - E) = A_1(x_{n-1} - E) + \cdots + A_{NA}(x_{n-NA} - E) + B_1 a_n + \cdots + B_{NB} a_{n-NB+1}$$

where

x_n is the observed value of the time series at time n,

NA is the number of autoregressive parameters, A_i,

NB is the number of moving average parameters, B_i,

E is the mean of the time series,

and

a_t is a series of independent random Standard Normal perturbations.

This is the form used in G05PAF. This is related to the form given in Section 3 by:

$B_1^2 = \sigma^2$,

$B_{i+1} = -\theta_i \sigma = -\theta_i B_1, \qquad i = 1, 2, \ldots, q$,

$NB = q + 1$,

$E = c$,

$A_i = \phi_i, \qquad i = 1, 2, \ldots, p$,

$NA = p$.

9 Example

This example calls G05PAF to set up the reference vector for an autoregressive model after initialization by G05KBF. The model is given by

$$x_t = 0.4x_{t-1} + 0.2x_{t-2} + \epsilon_t$$

where ϵ_t is a series of independent random Normal perturbations with variance 1.0. G05PAF is then called to generate a sample of ten observations, which are printed.

9.1 Program Text

```
*     G05PAF Example Program Text
*     Mark 20 Release. NAG Copyright 2001.
*     .. Parameters ..
      INTEGER          IP, IQ, N, NR
      PARAMETER        (IP=2,IQ=0,N=10,NR=IP+IQ+5+IP)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Local Scalars ..
      DOUBLE PRECISION AVAR, VAR, XMEAN
      INTEGER          I, IFAIL, IGEN, MODE
*     .. Local Arrays ..
      DOUBLE PRECISION PHI(IP), R(NR), THETA(1), X(N)
      INTEGER          ISEED(4)
*     .. External Subroutines ..
      EXTERNAL         G05KBF, G05PAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05PAF Example Program Results'
      WRITE (NOUT,*)
*     Set the ARMA model parameters
      XMEAN = 0.0D0
      PHI(1) = 0.4D0
      PHI(2) = 0.2D0
      AVAR = 1.0D0
*     Initialize the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 4234401
      ISEED(4) = 742355
*     IGEN identifies the stream.
      IGEN = 1
      CALL G05KBF(IGEN,ISEED)
      MODE = 0
      IFAIL = 1
*     Set up the reference vector
      CALL G05PAF(MODE,XMEAN,IP,PHI,IQ,THETA,AVAR,VAR,N,X,IGEN,ISEED,R,
     +            NR,IFAIL)
      IF (IFAIL.EQ.0) THEN
         MODE = 1
*        Generate a sample of 10 observations
         IFAIL = -1
         CALL G05PAF(MODE,XMEAN,IP,PHI,IQ,THETA,AVAR,VAR,N,X,IGEN,ISEED,
     +               R,NR,IFAIL)
         IF (IFAIL.EQ.0) THEN
            DO 20 I = 1, N
               WRITE (NOUT,99999) X(I)
   20       CONTINUE
         END IF
      ELSE
         WRITE (NOUT,99998) IFAIL
      END IF
*
99999 FORMAT (1X,F12.4)
99998 FORMAT (1X,' ** G05PAF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05PAF Example Program Results

      -1.0654
      -0.2828
      -2.0924
      -2.3304
      -2.5998
```

```
      -1.7143
      -2.4882
      -1.3882
      -2.2722
      -1.8806
```
